AFRL-HE-WP-TR-2000-0046



UNITED STATES AIR FORCE
RESEARCH LABORATORY

PHARMACOKINETIC MODELING OF JP-8 JET FUEL COMPONENTS I. NONANE AND C9-C12 ALIPHATIC COMPONENTS

Peter J. Robinson
MANTECH-GEO CENTER JOINT VENTURE
PO BOX 31009
DAYTON, OH 45437-0009

March 2000
Interim Report - October 1999 - March 2000

Human Effectiveness Directorate
Deployment and Sustainment Division
Operational Toxicology Branch
2856 G Street
Wright-Patterson AFB OH 45433-7400

NOTICES

When US Government drawings, specifications or other data are used for any purpose other than a definitely related Government procurement operation, the Government thereby incurs no responsibility nor any obligation whatsoever, and the fact that the Government may have formulated, furnished, or in any way supplied the said drawings, specifications, or other data is not to be regarded by implication or otherwise, as in any manner licensing the holder or any other person or corporation, or conveying any rights or permission to manufacture, use, or sell any patented invention that may in any way be related thereto.

Please do not request copies of this report from the Air Force Research Laboratory. Additional copies may be purchased from:

National Technical Information Service
5285 Port Royal Road
Springfield, Virginia 22161

Federal Government agencies and their contractors registered with the Defense Technical Information Center should direct requests for copies of this report to:

Defense Technical Information Service
8725 John J. Kingman Rd., Ste 0944
Ft. Belvoir, Virginia 22060-6218

DISCLAIMER

This Technical Report is published as received and has not
been edited by the Technical Editing Staff of the Air Force Research Laboratory.

TECHNICAL REVIEW AND APPROVAL

AFRL-HE-WP-TR-2000-0046

The animal use described in this study was conducted in accordance with the principles stated in the "Guide for the Care and Use of Laboratory Animals", National Research Council, 1996, and the Animal Welfare Act of 1966, as amended.

This report has been reviewed by the Office of Public Affairs (PA) and is releasable to the National Technical Information Service (NTIS). At NTIS, it will be available to the general public, including foreign nations.

This technical report has been reviewed and is approved for publication.

FOR THE DIRECTOR

David R. Mattie

DAVID R. MATTIE, Ph.D.
Acting Branch Chief, Operational Toxicology Branch
Air Force Research Laboratory

REPORT DOCUMENTATION PAGE	Form Approved OMB No. 0704-0188

1. AGENCY USE ONLY *(Leave blank)*	2. REPORT DATE	3. REPORT TYPE AND DATES COVERED
	March 2000	Interim Report - October 1999 - March 2000

4. TITLE AND SUBTITLE
Pharmacokinetic Modeling of JP-8 Jet Fuel Components I. Nonane and C9-C12 Aliphatic Components

5. FUNDING NUMBERS
Contract F41624-96-C-9010
PE 62202F
PR 7757
TA 7757A
WU 7757A114

6. AUTHOR(S)
Robinson, P.

7. PERFORMING ORGANIZATION NAME(S) AND ADDRESS(ES)
ManTech-Geo Center Joint Venture
PO Box 31009
Dayton, OH 45437-0009

8. PERFORMING ORGANIZATION REPORT NUMBER

9. SPONSORING/MONITORING AGENCY NAME(S) AND ADDRESS(ES)
Human Effectiveness Directorate
Air Force Research Laboratory
Wright-Patterson AFB, OH 45433-7400

10. SPONSORING/MONITORING AGENCY REPORT NUMBER
AFRL-HE-WP-TR-2000-0046

11. SUPPLEMENTARY NOTES

12a. DISTRIBUTION AVAILABILITY STATEMENT
Approved for public release; distribution is unlimited.

12b. DISTRIBUTION CODE

13. ABSTRACT *(Maximum 200 words)*

Exposures to JP-8 can occur from vapor, liquid or aerosol. Inhalation and dermal are the most prevalent routes of exposure, and acute effects include neurobehavioral deficits. Occupational exposures of interest include aircraft fuel tank workers and exhaust workers. JP-8 is a complex mixture of hundreds of components including straight chain alkanes, branched chain alkanes, cycloalkanes, diaromatics and napthalenes. A first step to developing a physiologically-based pharmacokinetic (PBPK) model for JP-8 is to develop models for representative components. Nonane is a good biomarker in breath for JP-8 aliphatic exposure ("fingerprint compound"). Nonane is a highly lipophilic compound (log Kow = 5.65), and has been observed to distribute preferentially in brain tissue. Its behavior in the body can be described in terms of a physiologically-based pharmacokinetic (PBPK) model that includes the blood, lungs, liver, muscle, and fat. Such a model is developed on the basis on in-house rat inhalation data (with measured blood levels), and validated by applying it to published rat inhalation data (blood and brain levels). The model is used to predict body burdens of nonane under occupation exposure conditions (ambient air concentrations for fuel tank workers and aircraft attendants) and is consistent with limited occupational body burden data (exhaled breath levels).

14. SUBJECT TERMS		15. NUMBER OF PAGES
Jet Fuel	Mixtures Toxicity	35
Nonane	Brain Exposure	16. PRICE CODE
PBPK Modeling		

17. SECURITY CLASSIFICATION OF REPORT	18. SECURITY CLASSIFICATION OF THIS PAGE	19. SECURITY CLASSIFICATION OF ABSTRACT	20. LIMITATION OF ABSTRACT
UNCLASSIFIED	UNCLASSIFIED	UNCLASSIFIED	UL

Standard Form 298 (Rev. 2-89) (EG)
Prescribed by ANSI Std. 239.18
Designed using Perform Pro, WHS/DIOR, Oct 94

THIS PAGE INTENTIONALLY LEFT BLANK.

PREFACE

This is the first in a series of technical reports describing the pharmacokinetic modeling of JP-8 fuel components. Specifically, this report outlines a physiologically-based pharmacokinetic (PBPK) model for nonane absorption and disposition in rats and humans, as an initial representative of C9 – C12 aliphatic hydrocarbons. In addition to nonane, models will be developed/assessed for other JP-8 components such as benzene, toluene, m-xylene and ethylbenzene. Finally, methods for combining compound-specific models in the risk assessment of complex mixtures such as JP-8, according to current guidelines published by the US EPA, will be developed.

This research was accomplished at the Operational Toxicology Branch, Human Effectiveness Directorate of the Air Force Research Laboratory.

TABLE OF CONTENTS

LIST OF FIGURES AND TABLES

ABSTRACT

Exposures to JP-8 can occur from vapor, liquid or aerosol. Inhalation and dermal are the most prevalent routes of exposure, and acute effects include neurobehavioral deficits. Occupational exposures of interest include aircraft fuel tank workers and exhaust workers. JP-8 is a complex mixture of hundreds of components including straight chain alkanes, branched chain alkanes, cycloalkanes, diaromatics and napthalenes. A first step to developing a physiologically-based pharmacokinetic (PBPK) model for JP-8 is to develop models for representative components. Nonane is a good biomarker in breath for JP8 aliphatic exposure ("fingerprint compound"). Nonane is a highly lipophilic compound (log Kow = 5.65), and has been observed to distribute preferentially in brain tissue. Its behavior in the body can be described in terms of a physiologically-based pharmacokinetic (PBPK) model that includes the blood, lungs, liver, muscle and fat. Such a model is developed on the basis of in-house rat inhalation data (with measured blood levels), and validated by applying it to published rat inhalation data (blood and brain levels). The model is used to predict body burdens of nonane under occupational exposure conditions (ambient air concentrations for fuel tank workers and aircraft attendants) and is consistent with limited occupational body burden data (exhaled breath levels).

INTRODUCTION

JP-8 is a complex mixture of hundreds of components (see Table 1). Component classes include: straight chain alkanes, branched chain alkanes, cycloalkanes, diaromatics and napthalenes. The Total Petroleum Hydrocarbon (TPH) Criteria Working Group has recently published a series of monographs outlining approaches for assessing the health impact of complex mixtures of hydrocarbons, including JP-8, in the environment (TPH, 1998). Central to the approach of this group is the notion of a relatively small number of fractions that are expected to behave similarly in the environment. These fractions consist of chemical components that have similar relevant physico-chemical properties such as:

- Equivalent carbon number
- Molecular weight
- Boiling point
- Vapor pressure
- Water solubility
- Octanol-water partition coefficient
- Henry's Law constant

Not all of these classification criteria are relevant for a chemical's behavior in the body (and additional parameters may be important), but this approach can form the basis for the classification of petroleum mixtures into a relatively small number of classes of similarly behaving chemicals. The TPH Working Group classified fractions into 6 aliphatic and 7 aromatic classes. Fractions relevant for JP-8 include the classes: aliphatics EC >6-8, EC >8-10, EC >10-12, EC >12-16, EC >16-21; and aromatics EC >7-8, , EC >8-10, EC >10-12, EC >12-16. In this classification, nonane can be considered representative of EC >8-10 aliphatics.

The specific details of such classification schemes are quite arbitrary, and other alternative classification schemes have been proposed. For example, the Massachusetts Department of Environmental Protection (MA DEP) has issued an interim final petroleum policy document that groups C9 – C17 alkanes/cycloalkanes together (for the purpose of determining a group RfD), and specifically identifies nonane as a reference compound for this group (MA DEP, 1994).

Table 1. Composition of JP-8 (Adapted from Total Hydrocarbon Working Group Series, Vol 3, Appendix A, 1997) (EC is equivalent carbon number)

Compound	Number of Carbons	EC	Weight Percent	Reference
Straight Chain Alkanes				
n-Heptane	7	7	0.03	API, 1993
n-Octane	8	8	0.9	API, 1993
n-Nonane	9	9	0.31	API, 1993
n-Decane	10	10	1.31	API, 1993
n-Undecane	11	11	4.13	API, 1993
n-Dodecane	12	12	4.72	API, 1993
n-Tridecane	13	13	4.43	API, 1993
n-Tetradecane	14	14	2.99	API, 1993
n-Pentadecane	15	15	1.61	API, 1993
n-Hexadecane	16	16	0.45	API, 1993
n-Heptadecane	17	17	0.08	API, 1993
n-Octadecane	18	18	0.02	API, 1993
Branched Chain Alkanes				
2,4,6-Thmethyiheptane	10		0.07	API, 1993
3-Methyloctane	9	8.78	0.04	API, 1993
2-Methyldecane	11		0.41	API, 1993
2,6-Dimethyldecane	12		0.66	API, 1993
2-Methylundecane	12		1.16	API, 1993
2,6-Dimethylundecane	13		2.06	API, 1993
Cycloalkanes				
1-Methyl-4-ethylcyclohexane	9		0.1	API, 1993
1,3,5-Trimethylcyclohexane	9		0.06	API, 1993
1,1,3-Thmethylcyclohexane	9	8.45	0.06	API, 1993
n-Butylcyclohexane	10		0.74	API, 1993
n-Propylcyclohexane	9		0.14	API, 1993
Hexylcyclohexane	12		0.93	API, 1993
Heptylcyclohexane	13		1	API, 1993
Straight Chain Alkenes				
Tridecene	14		0.73	API, 1993
Alkyl Benzenes				
o-Xylene	8	8.81	0.06	API, 1993
m-Xylene	8	8.6	0.06	API, 1993
1-Methyl-2-isopropylbenzene	10		0.56	API, 1993
1,3Dimethyl-5-ethylbenzene	10	10.51	0.62	API, 1993
1.2,4-Tnmethylbenzene	9	9.84	0.27	API, 1993
1,2,4-Tnethylbenzene	12	12.29	0.99	API, 1993
1,3,5-Tnethylbenzene	12	12.1	0.6	API, 1993
n-Heptylbenzene	13		0.25	API, 1993
n-Octylbenzene	14		0.61	API, 1993
Biphenyl	12		0.63	API, 1993
Phenylcyclohexane	12		0.87	API, 1993
Alkyl Naphthalenes				
Naphthalene	10	11.69	1.14	API, 1993
1-Methylnaphthalene	11	12.99	1.84	API, 1993
2-Methylnaphthalene	11	12.84	1.46	API, 1993

The US EPA has recently issued draft guidelines for the conduct of health risk assessments of chemical mixtures (EPA, 1999). These guidelines stress the importance of PBPK modeling in interpreting experimental data, stating that evaluation of risk posed by exposure to multiple chemicals can only come about with a solid understanding of the toxicity of chemical agents and the factors that control their absorption, metabolism, distribution and elimination. Further, the use of PBPK models, together with data from chemical interactions studies, could "form the very basis of mechanistic risk assessment methods for complex chemical mixtures" (EPA, 1999).

The purpose of this report is to begin such a process for JP-8 by describing a physiologically-based pharmacokinetic (PBPK) model for C9 – C12 or C9 – C17 aliphatic hydrocarbons. Nonane is chosen to be the representative compound for this group. Unpublished studies from our laboratory and elsewhere and literature derived information will be used to develop the nonane model. The utility of the model will be to

- Interpret biomarkers of exposure (exhaled breath, urine, blood measurements)
- Provide a tool for human dose-response assessment for JP-8 (assessing body burdens during occupational exposure)
- Provide a first "module" for a more comprehensive mixtures model for JP-8.

Nonane is a good biomarker in breath for JP8 aliphatic exposure. We now have some preliminary knowledge of ambient air levels of nonane under specific occupational settings, as well as body burden of nonane as reflected in pre- and post-exposure breath measurements of aircraft maintenance personnel (Pleil et al., in press). The PBPK model will be used as the basis for extrapolating to humans with the determination and substitution of appropriate species-specific parameters. This human-based model can take an input (exposure to ambient nonane) and predict an output (body burden of nonane), which can be compared with data.

Acute neurobehavioral effects have been observed as a result of exposure to JP-8 (Baldwin et al., 1998; Bhattacharya et al., 1998; Kaufman et al., 1998; see also Zeiger and Smith, 1998). In addition, nonane has been observed to distribute preferentially into rat brain tissue (Zahlsen et al., 1990). We thus include a brain compartment in the present model, so that we may have a future tool for predicting brain levels of nonane.

Nonane Model

Data

Experimental data consists of a series of inhalation experiments with Fischer 344 rats at 100, 500 and 1000 ppm. Exposure was via nose only. Movement of the animals was restricted. Experimental details are given elsewhere (see Caracci et al., Technical Report, in preparation). The results are summarized in figure 1 below.

Figure 1. Rat (nose only) inhalation data for 3 nonane concentrations. Exposure was for 4 hr, and blood concentrations were followed for a further 4 hr.



Partition coefficients were measured by vial equilibration methods (Gargas et al., 1989)

Model

The PBPK model for nonane is based on the Ramsay-Anderson perfusion-limited inhalation exposure model (Ramsay & Andersen, 1984) illustrated in Figure 2. The nonane model consists of lung, liver, brain, slowly perfused tissue (muscle), and fat compartments.

Figure 2. Schematic representation of the PBPK model for nonane inhalation.



Model Parameters

Physiological parameters for the rat are given in Table 2 (see Brown et al., 1997). The only additional parameters used in the fitting were the metabolic parameters, Vmax and Km for the liver. Gas uptake studies to specifically determine metabolic parameters were not performed. However, in general, the results were not particularly sensitive to the specific values taken by these two parameters.

Table 2. Physiological parameters for nonane model

	Rat	*Human*
Body Weight (kg)	0.3	70
Cardiac Output (L/hr-kg)	19	4.46
Alveolar Ventilation (L/hr-kg)	19	6.43
Fractional blood flow to liver	0.25	0.23
Fractional blood flow to fat	0.09	0.09
Fraction liver tissue	0.04	0.026
Fraction fat tissue	0.09	0.213
Volume of blood in alveolar region	0.0005	0.0005

Measured tissue/air partition coefficients for rat blood, liver, fat, muscle and brain (Unpublished data from this laboratory) are given in Table 3.

Table 3. Measured tissue/Air partition coefficients for rat (400ppm)

	Blood	Liver	Fat	Muscle	Brain
Mean	5.13	6.64	1254	7.13	25.85
SD	1.38 (n=15)	1.94 (n=24)	372 (n=15)	1.32 (n=15)	6.83 (n=10)

Results

Rat

Data Fitting

The PBPK model described above (without the brain compartment) was manually fitted to the mean blood concentrations for each of the three exposure concentrations (Figures 3, 4 and 5). The only adjustable parameters in the fitting were the metabolic parameters Vmax and Km. These metabolic parameters had a relatively minor impact on the overall fit of the curves. Overall, it was found that a Vmax of around 10 mg/kg-hr and a Km of 0.1 mg/L gave the most reasonable fits over the 3 data sets. Note that except for the lowest blood concentrations at 100 ppm exposure (Figure 3), metabolism was nearly always saturated.

Figure 3. Experimental data (same as in Figure 1) and model curve for rats breathing 100 ppm nonane (nose only from 0 to 4 hours.) (means, n=9). In this case, Vmax = 10 mg/h and Km = 0.1 mg/L.



Figure 4. Experimental data (same as in Figure 1) and model curve for rats breathing 500 ppm nonane (nose only) from 0 to 4 hours (means, n=10). Again, Vmax = 10 mg/h and Km = 0.1 mg/L. The poor fit during exposure is likely due to breathing difficulties with the experimental apparatus, and hence a reduced alveolar ventilation rate (see text).



The fit of the 500 ppm data (Figure 4) was significantly worse than the other two; the model consistently overestimated the blood concentrations during the exposure period (0 – 4 hr). This fit could not be improved by adjusting the metabolic parameters. However, reducing the alveolar ventilation rate by some 50% (from 19 to 10 L/hr-kg) significantly improved the fit

to the data. This is consistent with the observation in some of the experiments that breathing was observed to be somewhat hampered by the experimental apparatus.

All theoretical curves failed to capture the nuances of the apparent shape of the experimental data at late time points after the end of exposure (although there is a great deal of scatter in the data – see Figure 1). This is likely due to a lack of detail in the model in relation to distribution into body fats – see the *Discussion* section below.

Figure 5. Experimental data (same as in Figure 1) and model curve for rats breathing 1000 ppm nonane (nose only from 0 to 4 hours.) (means, n=9). Again, Vmax = 10 mg/h and Km = 0.1 mg/L.



Validation

The model was assessed and validated by applying it to published data (Zahlsen et al., 1990 & 1992) without any further parameter adjustments. These investigators measured blood and tissue concentrations of nonane (and other compounds) in male Sprague-Dawley rats immediately following 12 h daily exposure to 100 ppm and 1000 ppm over a period of 14 days. Although the data points are sparse, model predictions are in agreement with the data at both 1000 and 100 ppm (Figures 6 and 7).

Figure 6. Model prediction of rat blood nonane concentration for the exposure scenario of Zahlsen et al., 1990 (1000 ppm, 12 h daily exposure for up to 14 days). Data are adapted from Figure 1 of that paper. In this and the following figures, predicted curves were obtained using previously fitted (Vmax, Km) together with the rat physiological and chemical parameters given in Tables 2 and 3 .



Figure 7. Model prediction of rat blood nonane concentration for the exposure scenario of Zahlsen et al., 1992 (100 ppm, 12 h daily exposure for 3 days), using measured and previously fitted model parameter values. Data are adapted from Table 2 of that paper. Experimental errors were reported around 10 – 15%.



Discussion

Limitations of the model

Nonane is metabolized in the rat at relatively high rates to hydroxyl derivatives prior to conversion into the corresponding keto form, as determined with a cytochrome p450 containing mixed function oxidase system (Clayton et al., 1982). However, separate gas uptake studies to specifically determine the metabolic parameters (Vmax and Km) were not conducted. As a result, these parameters were fitted to the pharmacokinetic data.

A second limitation of the data set on which the model is based is that the alveolar ventilation rate was not measured under the specific exposure conditions of the experiment (immobilized animals fitted with nose cones for nose-only exposure). It is likely that the ventilation rate is much lower than "normal" values reported for rats.

Figures 3-5 show fairly reasonable agreement between the model and the mean data values, in the light of inter-animal variability (see error bars in Figure 1). However, there are some obvious discrepancies. Firstly, Figure 4 shows that the model over-predicts the blood concentration of nonane during the 4 h absorption phase by about 50 – 100%. This can not be satisfactorily removed by adjusting the metabolic parameters Vmax and Km. It can be removed by reducing the alveolar ventilation rate from 19 L/kg/h to around 10 L/kg/h, perhaps corresponding to breathing difficulties with the inhalation apparatus. Overestimation of the air-blood partition coefficient could also explain this discrepancy, but this is unlikely to be the case in just one of the three exposure groups.

Secondly, all sets of data show a systematic underprediction of the slope of the terminal phase from around 4.5 h (30 min after cessation of exposure). In the model, this phase is determined by the slow release of the highly lipophilic nonane from the single fat compartment (characterized by the measured fat-air and blood-air partition coefficients).It appears that in reality the fat is effectively replenishing the nonane in the blood, and being itself depleted, at a significantly greater rate than predicted by the model and these measured partition coefficients. There are a number of alternative (though related) explanations for this behavior:

- The measured fat partition coefficient is an overestimate of its "true" value. This can be because the sampled fat is not representative of the fat compartment of the animal as a whole, or because the fat compartment is highly heterogeneous in terms of its partition coefficient.

- The model oversimplifies the fat as a single (homogeneous) compartment.

- There are additional compartments (not in the model) that influence the behavior of the blood concentrations at these late times.

- The fat volume has been overestimated in the model

These notions can be generalized. Our PBPK model oversimplifies by incorporating only a small number of (homogeneous) compartments (the fat, muscle, liver and brain). In reality, there are a much larger number of "sub-compartments", even within a single tissue such as the fat and muscle. Each (arbitrary) sub-compartment has its own partition coefficient that differs slightly from its neighbors, so that instead of a few compartments with a small number of widely separated partition coefficients, we really have more like a continuum of such coefficients. This would result in a smoothing out of the "biphasic" disappearance curves of Figures 3-5, rather like what is actually observed. Similar heterogeneities are indicated by the tendency of the data towards a slower, smoother rise than predicted during the absorption phase.

Extrapolation to Human Exposures

At present we do not have tissue/air partition coefficients for human blood or other tissues. However, we may make extrapolations to human exposure conditions with limited confidence using human physiological values and rat partition coefficients, bearing in mind that this analysis will be modified once human partition data becomes available.

Figure 8 shows the model predictions for the blood values of nonane for two hypothetical human exposure scenarios corresponding to fuel tank entry workers and attendants. Ambient air measurements by Pleil et al. (in press) give nonane concentrations of around 34 and 1.8 ppb. In these simulations, exposures were assumed to be for 4 h (no specific exposure durations were given). Measured exhaled breath before and at some undefined period (shortly?) following exposure were used to calculate blood concentrations of nonane in these workers (using the blood-air partition coefficient for rat blood). These calculated blood values (and their uncertainties) are given as the horizontal boxes in Figure 8. Figure 8b also shows the background nonane levels in people not occupationally exposed to JP-8 (horizontal arrow).

Given the very large uncertainties in the exposures and the calculations, the agreement of the predicted and measured blood levels for the aircraft attendants (1.8 ppb ambient nonane) both before and after exposure is quite reasonable. Blood levels following exposure are if anything overestimated by the model. This may be due to the fact that volatile hydrocarbons such as nonane have a greater affinity for rat red blood cells than human red blood cells (Lam et al.,

1990), leading to an overestimate of the blood-air partition coefficient (and hence inhalation uptake) for humans based on rat partition data.

Since these workers are likely exposed on a more or less daily basis, blood levels before exposure would be expected to approximate levels predicted some 20 hr following exposure, as is indeed the case.

On the other hand, blood levels for the tank entry attendants are greatly overestimated by the model. This is likely due to the wearing of respirators by these workers. (Attendants do not wear respirators). If this is indeed the case, respirators seem to reduce blood levels of nonane by almost 3 orders of magnitude (or more, if we allow for the likelihood that dermal exposure likely contributes significantly to blood levels in these individuals). A comparison of dermal and inhalation exposure to JP-8 components will be assessed separately.

Figure 8. Predicted nonane concentrations in human blood following 4 h exposure to 34 (a) and 1.8 (b) ppb ambient nonane in air to workers (tank entry and attendant) (using rat partition coefficients). Boxes represent range of measured values from Pleil et al., in press.



(a)



(b)

Brain and Tissue concentrations

Zahlsen et al (1990, 1992) noted that the brain concentration of nonane in their rats was surprisingly high compared with the other compounds they studied. These investigators looked at brain concentrations at various times during the course of repeated daily exposures to 100 and 1000 ppm nonane in separate experiments (Zahlsen et al., 1990 & 1992). The present model (including the brain compartment with its measured tissue-air partition coefficient given in Table 3) was compared with these data without any parameter adjustment, and was found to fit remarkably well (Figures 9 and 10).

Figure 9. Model prediction of rat brain nonane concentration for the exposure scenario of Zahlsen et al., 1990 (1000 ppm, 12 h daily exposure for up to14 days), using measured and previously fitted parameter values. Data are adapted from Figure 2 of that paper.



Figure 10. Model prediction of rat brain nonane concentration for the exposure scenario of Zahlsen et al., 1992 (100 ppm, 12 h daily exposure for 3 days), using measured and previously fitted parameter values. Data are adapted from Table 2 of that paper.



Human brain concentrations can also be predicted. This is useful in light of the observation of behavioral effects resulting from acute exposures to JP-8 (Baldwin et al., 1998; Bhattacharya et al., 1998; Kaufman et al., 1998). In the case of aircraft attendant exposure (see above), the model appears to describe blood data reasonably well (Figure 8b). For these same exposure conditions (4 hours at an ambient concentration of 1.8 ppb), predicted brain tissue concentrations of nonane are given in Figure 11. The model can be used to predict brain concentrations under various exposure scenarios, ultimately to be compared with possible neurobehavioral effects.

Figure 11. Predicted brain nonane concentrations for humans during and following 4 hr. exposure at 1.8 ppb (using partition coefficients measured for rat blood and tissue).



Conclusions

The current model was developed based on existing in-house data. These data have two important limitations: separate gas uptake studies to specifically determine the metabolic parameters (Vmax and Km) were not conducted, and alveolar ventilation rates for the specific experimental conditions were not determined. (These data gaps remain to be filled). Standard values were used for the latter, while the former were fitted to inhalation data at three concentrations. The extrapolation of this model to published rat data, however, was quite successful. Both blood and brain concentrations were predicted (without any further parameter adjustment) that were close to experimental values (Figures 6,7 and 9,10) over a 10-fold range of inhaled air concentrations.

The applicability of such a simple, non-specific model for nonane suggests that (except for metabolism) the main processes defining the accumulation and disappearance of nonane from the tissues of the body are primarily also non-specific and applicable to a wider range of aliphatic hydrocarbons. (There is nothing really unusual going on here from a chemical-specific point of view). This supports our notion of nonane as a representative compound for a wider range of aliphatics (such as C9 – C12 or C9 – C17). Once estimates of the relevant chemical-specific parameters (partition coefficients) for other members of this class are made (together with studies of overall metabolism), their behavior in the body is likely to broadly follow that proscribed in the present model.

Having an idea of how individual compounds behave in the body does not of course explain how they will behave in concert. In assessing the behavior of complex mixtures such as JP-8, we not only need to know how members of the same class behave together, but also how different chemical classes may interact with each other. Pharmcokinetic analysis in general and PBPK models in particular are likely to provide some insight into this issue, as recently suggested by the EPA (1999) in their *Guidance for Conducting Health Risk Assessment of Chemical Mixtures.* However, the specifics of the form such analyses are likely to take are not yet clear. In the case of major hydrocarbon components of JP-8, a likely site of interaction is hepatic metabolism by the 2E1 isozyme of cytochrome P-450 (CYP2E1). In such a case, a PBPK model may be constructed in which models for individual components are linked via this common pathway. Tardif et al. (1997) successfully developed such a model for a ternary mixture of toluene, m-xylene and ethylbenzene (all of which are components of JP-8 actually observed in exhaled breath of fuels workers), in which there is competitive inhibition of this particular isoenzyme.

Acknowledgements

The author would like to thank Jim McDougal, Allen Vinegar, John Frazier and Dan Pollard for valuable comments and discussions.

References

API (1993). Petroleum product Surveys. American Petroleum Institute, Washington, D.C.

Baldwin CM, Podgornik MN, Young RS, Rao G, Houston FP, Barnes CA, and Witten ML. Biobehavioral And Memory Alterations Related To Moderate And High Dose JP-8 Jet Fuel Exposure. 1998 International Conference on The Environmental Health and Safety of Jet Fuel, San Antonio, TX, 1-3 April, 1998.

Baldwin, C.M., Michelle N. Podgornik, Frank P. Houston, R. Scott Young, Carol A. Barnes, Mark L. Witten. JP-8 jet fuel may affect arousal rather than cognitive function: future neurobehavioral research directions. AFOSR JP-8 Jet Fuel Toxicology Workshop, The University of Arizona, Tucson, Arizona, December 2-3, 1998

Bhattacharya, A. and Smith, L., Effect of Jet Fuel and Physical Risk Factors on Ability to Maintain Postural Balance During Task Performance. 1998 International Conference on The Environmental Health and Safety of Jet Fuel, San Antonio, TX, 1-3 April, 1998.

Brown, R.P., M.D. Delp, S.L. Lindstedt, L.R. Rhomberg and R.P. Beliles. Physiological parameter values for physiologically based pharmacokinetic models. Tox and Ind. Health 13, 407-484, 1997.

Caracci, M., Weisman, W, J. McDougal, D. Pollard, J. McCafferty and E. Merrill. A study in female Fischer-344 rats to determine pharmacokinetics for n-nonane. AFRL/HEST Technical Report (in preparation).

Clayton, G. D. and F. E. Clayton (eds.). Patty's Industrial Hygiene and Toxicology: Volume 2A, 2B, 2C: Toxicology. 3rd ed. New York: John Wiley Sons, 1981-1982. 3191

EPA/NCEA Guidance for Conducting Health Risk Assessment of Chemical Mixtures (April, 1999, NCEA-C-0148) 1999.

Gargas, M.L., Burgess, R.J., Voisard, D.E., Cason, G.H. and Andersen, M.E.. Partition coefficients of low molecular weight volatile chemicals in various liquids and tissues. TAP 98, 87-99, 1989.

Giardino, N. Using PBPK modeling to guard worker safety and perform exposure and dose reconstruction: implications for risk. Proc. International Conference on the Environmental Health and Safety of Jet Fuel. San Antonio, TX April 1998.

Kaufman L, Bhattacharya A, Dunn D, Lockey J, Succop P, Lemasters G. Preliminary Results Of Occupational Exposure To Jet Fuel On Hearing. 1998 International Conference on The Environmental Health and Safety of Jet Fuel, San Antonio, TX, 1-3 April, 1998.

Lam, C-W, TJ Galen JF Boyd DL Pierson. Mechanism of transport and distribution of organic solvents in blood. TAP 104, 117-129, 1990.

M. A. Medinsky et al. Benzene Risk Assessment: Transgenic Animals and Human Variability CIIT Activities (Vol.17, No.7).

Massachusetts Department of Environmental Protection. Interim final petroleum policy: development of health-based alternative to the total petroleum hydrocarbon (TPH) parameter. June 1994.

Pleil, J.D., L.B. Smith & S.D. Zelnick. Personal exposure to JP-8 jet fuel vapor and exhaust at air force bases (draft report).

Ramsay, JR & Andersen, ME. A physiologically based description of the inhalation based pharmacokinetics of styrene in rats and humans. TAP, 73, 159-175, 1984.

Tardif, R., Charest-Tardif, G., Brodeur, J. and K. Krishnan. Physiologically based pharmacokinetic modeling of a ternary mixture of alkyl benzenes in rats and humans. TAP 144, 120-134, 1997.

TPH Working Group Series (Full text available online at: http://www.aehs.com/). Volume 1: Petroleum Hydrocarbon Analysis of Soil and Water in the Environment. Wade Weisman 1998.

TPH Volume 2: Composition of Petroleum Mixtures. Thomas L. Potter; Kathleen E. Simmons. 1998.

TPH Volume 3: Selection of Representative TPH Fractions Based on Fate and Transport Considerations. John Gustafson; E: Joan Griffith Tell; E: Doug Orem. 1998

TPH Volume 4: Development of Fraction Specific Reference Doses (RfD's) and Reference Concentration (RfC's) for Total Petroleum Hydrocarbons. D.A. Edwards; E: M.D. Amoroso; E: A.C. Tummey; E: S.H. Youngren; E: C.J. Bevan; E: A. Tveit; E: L.A. Hayas; E: D.V. Nakies. 1998.

Volume 5: Human Health Risk-Based Evaluation of Petroleum Contaminated Sites: Implementation of the Working Group Approach. Donna Vorhees and Wade Weisman. 1998.

Zahlsen, K., A.M. Nilsen, I. Eide and O.G. Nilsen. Accumulation and distribution of aliphatic (n-nonane), aromatic (1,2,4 trimethylbenzene) and napthenic (1,2,4 trimethylcyclohexane) hydrocarbons in the rat after repeated inhalation. Pharmacology & Toxicology 67, 436-440, 1990.

Zahlsen, K., I. Eide A.M. Nilsen, and O.G. Nilsen. Inhalation kinetics of C6 to C10 aliphatic, aromatic and napthenic hydrocarbons in rat after repeated exposures. Pharmacology & Toxicology 71, 144-149, 1992.

Zeiger, E. and Smith, L. The First International Conference on The Environmental Health and Safety of Jet Fuel. Environmental health Perspectives, 106, 763-764, 1998.

Appendix: Representative ACSL Code for PBPK Model for Nonane

The following are the .csl and .cmd files for simulating rat exposure to 100 ppm nonane.

```
PROGRAM: non100.CSL PHYSIOLOGICAL MODEL
'INHALED LIPOPHILIC CHEMICAL MODEL WITH FLOW-LIMITED TISSUES'
'Modified from CSU PB-PK COURSE AUGUST 1994'
'PREPARED BY HARVEY CLEWELL, Modified by Peter Robinson, 9/30/99'

INITIAL

'Physiological Parameters ***********************************'
CONSTANT      BW = 0.3     $'Body weight (kg)'
CONSTANT     QCC = 19.     $'Cardiac output (L/hr-1kg)'
CONSTANT     QPC = 19.     $'Alveolar ventilation (L/hr-1kg)'
CONSTANT     QLC = 0.25    $'Fractional blood flow to liver'
CONSTANT     QFC = 0.09    $'Fractional blood flow to fat'
CONSTANT     VLC = 0.04    $'Fraction liver tissue'
CONSTANT     VFC = 0.09    $'Fraction fat tissue'
CONSTANT     VBC = 0.0005 $'Volume of blood in alveolar region'

'Chemical specific parameters ******************************'

CONSTANT      PB = 5.134   $'Blood/air partition coefficient'
CONSTANT      PL = 1.29    $'Liver/blood partition coefficient'
CONSTANT      PF = 244.    $'Fat/blood partition coefficient'
CONSTANT      PT = 1.43    $'Slowly perfused tissue/blood partition (muscle)'
CONSTANT      MW = 128.    $'Molecular weight (g/mol)'
CONSTANT   VMAXC = 10.     $'Capacity of saturable metabolism (mg/hr)'
CONSTANT      KM = 0.1     $'Affinity of saturable metabolism (mg/L)'

'Calculated parameters'

      VF = VFC*BW                        $'Volume of fat - L'
      VL = VLC*BW                        $'Liver volume - L'
      VT = 0.91*BW-VL-VF                 $'Tissue volume - l'
      QC = QCC*BW**0.74                  $'Cardiac output - L/hr'
      QP = QPC*BW**0.74                  $'Alveolar ventilation-L/hr'
      QL = QLC*QC                        $'Liver blood flow - L/hr'
      QF = QFC*QC                        $'Fat blood flow - L/hr'
      QT = QC-QF-QL                      $'Tissue blood flow L/hr'
      VB = VBC*BW                        $'Blood volume - L'
    VMAX = VMAXC*BW**0.7                 $'Metabolic capacity - mg/hr'

'Parameters for simulation of experiment**********************'

CONSTANT    CONC = 100.   $'Concentration in air (ppm)'

             CIX = CONC*MW/24450.        $'Concentration in mg/L'

     'Timing commands'
```

```
CONSTANT   TSTOP = 8.    $'Length of experiment (hrs)'
CONSTANT  POINTS = 240.  $'Number of points in plot'
CONSTANT  LENGTH = 4.    $'Duration of exposure (hrs)'

      'Exposure Definitions'

    CINT  = TSTOP/POINTS  $'Communication interval'


     ' Method of integration'

ALGORITHM IALG = 2         $'Gear method for stiff systems'

END     $'End of initial'

DYNAMIC

'-------------------------------------------------------'

DISCRETE CAT1 $'Schedule events to turn exposure on and off daily'
   INTERVAL CAT = 1000.   $'Set interval larger than any TSTOP'
                           '(to prevent multiple exposure)'
            CI = CIX              $'Start inhalation exposure'
   SCHEDULE CAT2 .AT. T + LENGTH   $'Schedule end of exposure'
END $'of CAT1'

DISCRETE CAT2
            CI = 0.               $'End inhalation exposure'
END $'of CAT2'
'-------------------------------------------------------'

DERIVATIVE

      'AT = Amount in tissues (mg)'
      RAT = QT*(CA-CT/PT)
       AT = INTEG(RAT,0.)
       CT = AT/VT

      'AF = Amount in fat tissue (mg)'
      RAF = QF*(CA-CF/PF)
       AF = INTEG(RAF,0.)
       CF = AF/VF

      'AL = Amount in liver tissue (mg)'
      RAL = QL*(CA-CL/PL)-RAM
       AL = INTEG(RAL,0.)
       CL = AL/VL

      'AM = Amount metabolized (mg)'
      RAM = VMAX*CL/PL/(KM+CL/PL)
       AM = INTEG(RAM,0.)

      'CA = Blood concentration (mg/L)'
      RVB = QL*CL/PL+QT*CT/PT+QF*CF/PF
       CV = RVB/QC
      RAB = QC*CV+QP*CI-QP*CX-QC*CA      !blood transport + gas exchange
       AB = INTEG(RAB,0.)
       CA = AB/VB
```

```
      CX = CA/PB
   CXPPM = (0.7*CX+0.3*CI)*24450./MW

  'TMASS = mass balance (mg)'
   TMASS = AF+AL+AT+AM+AB

 TERMT(T.GE.TSTOP)

END       $'End of derivative'
END       $'End of dynamic'
END       $'End of program'
```

```
  'FILE INHAL.CMD FOR CSU PB-PK COURSE AUGUST 1994'
  'PREPARED BY HARVEY CLEWELL,Modified by Peter Robinson
    '


SET TITLE ='VOLATILE COMPOUND TIME-COURSE                    '
PREPAR T,CL,CF,TMASS,CA,CV,CXPPM,RAM,AM
OUTPUT T,CV,'NCIOUT'=10

PROCED RAT
SET BW=0.3,VFC=0.09
END

DATA RAT100 (t,cv)
.5      0.427
1.      0.523
2.      0.516
3.      0.676
4.      0.891
4.08    0.430
4.16    0.362
4.25    0.304
4.5     0.205
5       0.114
6       0.037
7       0.030
8       0.019
END

PROCED PL
SET TITLE ='NONANE INHALATION IN RATS - 100 PPM, NOSE ONLY 2-22-96, 5-3-9
6 (Means, n=9)'
PLOT /DATA=RAT100 CV /LOG /LO=.01 /HI=10 ...
     /TAG=' - Blood Concentration (mg/L)'...
     /XTAG = ' - Time (hrs)' /COLOR=0 /XHI=8 /CHAR=2
END

PROCED HP
SET DEVPLT=4
FILE/PLTFILE='NON.hgl'
PLOT /DATA=RAT100 CV /LOG /LO=.01 /HI=10 ...
END
```

```
PROCED SHOWIT $ 'Show values of key parameters'
D QPC,QCC,QFC,QLC
D BW,VFC,VLC
D PB,PF,PL,PT
D MW,VMAXC,KM
D CONC
D LENGTH,TSTOP,POINTS
END

PROCED ONCE
SET NRWITG=.F.
END

PROCED OVER
SET NRWITG=.T.
END

SET FTSPLT=.T.,XINCPL=5.,SYMCPL=.F.,GRDCPL=.F.,WESITG=.F.
SET HVDPRN=.T.
```

The following are the .csl and .cmd files for simulating human exposure to 1.8 ppm nonane. This model also includes the brain compartment.

```
PROGRAM: Brain.CSL PHYSIOLOGICAL MODEL
'INHALED LIPOPHILIC CHEMICAL MODEL WITH FLOW-LIMITED TISSUES PLUS BRAIN'
'Modified from CSU PB-PK COURSE AUGUST 1994'
'PREPARED BY HARVEY CLEWELL, Modified by Peter Robinson, 10/1/99'

INITIAL

'Physiological Parameters ***********************************'

CONSTANT      BW = 70     $'Body weight (kg)'
CONSTANT     QCC = 4.46   $'Cardiac output (L/hr-1kg) (ILSI p. 42)'
CONSTANT     QPC = 6.43   $'Alveolar ventilation (L/hr-1kg) (ILSI p 77)'
CONSTANT     QLC = 0.23   $'Fractional blood flow to liver (ILSI)'
CONSTANT     QFC = 0.09   $'Fractional blood flow to fat'
CONSTANT     VLC = 0.026  $'Fraction liver tissue (ILSI)'
CONSTANT     VFC = 0.213  $'Fraction fat tissue (ICRP, ILSI p. 22)'
CONSTANT     VBC = 0.0005 $'Volume of blood in alveolar region'
CONSTANT     QBRC = 0.115   $'Fractional blood flow to brain (ILSI p. 51)'
CONSTANT     VBRC = 0.02   $'Fraction brain tissue'

'Chemical specific parameters *********************************'

CONSTANT      PB = 5.134  $'Blood/air partition coefficient'
CONSTANT      PL = 1.29   $'Liver/blood partition coefficient'
CONSTANT      PF = 244.   $'Fat/blood partition coefficient'
CONSTANT      PT = 1.43   $'Slowly perfused tissue/blood partition (muscle)'
CONSTANT      MW = 128.   $'Molecular weight (g/mol)'
CONSTANT   VMAXC = 10.    $'Capacity of saturable metabolism (mg/hr)'
CONSTANT      KM = 0.1    $'Affinity of saturable metabolism (mg/L)'
CONSTANT      PBR = 5.03  $'Brain/blood partition coefficient'

'Calculated parameters'

      VF = VFC*BW                      $'Volume of fat - L'
      VL = VLC*BW                      $'Liver volume - L'
      VT = 0.91*BW-VL-VF               $'Tissue volume - l'
      QC = QCC*BW**0.74                $'Cardiac output - L/hr'
      QP = QPC*BW**0.74                $'Alveolar ventilation - L/hr'
      QL = QLC*QC                      $'Liver blood flow - L/hr'
      QF = QFC*QC                      $'Fat blood flow - L/hr'
      QT = QC-QF-QL                    $'Tissue blood flow L/hr'
      VB = VBC*BW                      $'Blood volume - L'
    VMAX = VMAXC*BW**0.7               $'Metabolic capacity - mg/hr'
     VBR = VBRC*BW                $'Volume of brain - L'
     QBR = QBRC*BW                $'Brain blood flow L/hr'

'Parameters for simulation of experiment**********************'

CONSTANT    CONC = 1.8   $'Concentration in air (ppm)'

            CIX = CONC*MW/24450.        $'Concentration in mg/L'

     'Timing commands'

CONSTANT   TSTOP = 24.   $'Length of experiment (hrs)'
CONSTANT  POINTS = 1000.  $'Number of points in plot'
CONSTANT  LENGTH = 4.    $'Duration of exposure (hrs)'
```

```
      'Exposure Definitions'

    CINT  = TSTOP/POINTS  $'Communication interval'


     ' Method of integration'

ALGORITHM IALG = 2          $'Gear method for stiff systems'

END     $'End of initial'

DYNAMIC

'-------------------------------------------------------'

DISCRETE CAT1 $'Schedule events to turn exposure on and off daily'
   INTERVAL CAT = 24.    $'Set interval larger than any TSTOP'
                          '(to prevent multiple exposure)'
            CI = CIX                $'Start inhalation exposure'
   SCHEDULE CAT2 .AT. T + LENGTH    $'Schedule end of exposure'
END $'of CAT1'

DISCRETE CAT2
            CI = 0.                 $'End inhalation exposure'
END $'of CAT2'
'-------------------------------------------------------'

DERIVATIVE

      'AT = Amount in tissues (mg)'
      RAT = QT*(CA-CT/PT)
       AT = INTEG(RAT,0.)
       CT = AT/VT

      'AF = Amount in fat tissue (mg)'
      RAF = QF*(CA-CF/PF)
       AF = INTEG(RAF,0.)
       CF = AF/VF

      'ABR = Amount in brain tissue  (mg)'
      RABR = QBR*(CA-CBR/PBR)
       ABR = INTEG(RABR,0.)
       CBR = ABR/VBR

      'AL = Amount in liver tissue (mg)'
      RAL = QL*(CA-CL/PL)-RAM
       AL = INTEG(RAL,0.)
       CL = AL/VL

      'AM = Amount metabolized (mg)'
      RAM = VMAX*CL/PL/(KM+CL/PL)
       AM = INTEG(RAM,0.)

      'CA = Blood concentration (mg/L)'
      RVB = QL*CL/PL+QT*CT/PT+QF*CF/PF
       CV = RVB/QC
      RAB = QC*CV+QP*CI-QP*CX-QC*CA      !blood transport + gas exchange
       AB = INTEG(RAB,0.)
       CA = AB/VB
       CX = CA/PB
    CXPPM = (0.7*CX+0.3*CI)*24450./MW

   'TMASS = mass balance (mg)'
```

```
    TMASS = AF+AL+AT+AM+AB+ABR

  TERMT(T.GE.TSTOP)

END       $'End of derivative'
END       $'End of dynamic'
END       $'End of program'
```

```
   'FILE INHAL.CMD FOR CSU PB-PK COURSE AUGUST 1994'
   'PREPARED BY HARVEY CLEWELL,Modified by Peter Robinson

SET TITLE ='VOLATILE COMPOUND TIME-COURSE
PREPAR T,CL,CF,TMASS,CA,CV,CXPPM,RAM,AM,CBR
OUTPUT T,CA,CF,'NCIOUT'=10

DATA HUMAN (t,cv)

END

PROCED PL
SET TITLE ='NONANE INHALATION IN HUMANS'
PLOT /DATA=HUMAN CBR /LOG /LO=.00001 /HI=1 ...
     /TAG=' - Brain Concentration (mg/L)'...
     /XTAG = ' - Time (hrs)' /COLOR=0 /XHI=24 /CHAR=2
END

PROCED SHOWIT $ 'Show values of key parameters'
D QPC,QCC,QFC,QLC
D BW,VFC,VLC
D PB,PF,PL,PT
D MW,VMAXC,KM
D CONC
D LENGTH,TSTOP,POINTS
END

PROCED ONCE
SET NRWITG=.F.
END

PROCED OVER
SET NRWITG=.T.
END

SET FTSPLT=.T.,XINCPL=5.,SYMCPL=.F.,GRDCPL=.F.,WESITG=.F.
SET HVDPRN=.T.
```